
Abstract. This paper proposes modeling laws for Alloy, a formal object-
oriented modeling language. These laws are important not only to define the
axiomatic semantics of Alloy but also to guide and formalize popular software
development practices. In particular, these laws can be used to formaly refac-
tor specifications. As an example, we formally refactor a specification for Java
types.

1. Introduction

Laws of programming are important not only to define the axiomatic semantics of pro-
gramming languages but also to assist the software development process. In fact, be-
sides being used as a foundation for informal development practices like refactoring,
which is becoming popular due to modern methodologies like Extreme Programming
[Beck, 2000], these laws are very helpful to motivate the practical use of formal develop-
ment methods.

Although they have not been sufficiently studied yet, modeling laws, or laws of
design models, might bring similar benefits, but with a greater impact on reliability and
productivity, since they are used in earlier stages of the software development process.
In order to explore that, this paper introduces and formalizes modeling laws. We also
show how those laws can be used to refactor Alloy specifications. In particular, we focus
on transformations of project models in Alloy [Jackson et al., 2001], a formal object-
oriented modeling language.

We chose Alloy rather than the Unified Modeling Language
(UML) [Booch et al., 1999] and the Object Constraint Language (OCL)
[Kleppe and Warmer, 1999], because Alloy has a simpler semantics and is not am-
biguous, allowing then a consistent formalization. Another advantage of Alloy is the
support for automatic specification analysis, which is an important aspect for using for-
mal methods in practice. Furthermore, despite its simple semantics, Alloy is expressive
enough to model a great variety of applications. Probably, our laws can also be useful
for reasoning about UML models if we give the semantics of UML class diagrams using
Alloy.

Besides clarifying aspects of Alloy’s semantics, these basic laws serve as a basis
for deriving more elaborate laws for practical applications of model transformations, such
as introducing design patterns to a model. We will not show here due to the lack of
space. Since we proved that the basic laws preserve semantics, we assure that these more
elaborate laws derived, using the basic laws, also preserve semantics. The focus of these
basic laws is to use them to derive more elaborate laws that restructure models for which

we want to assure that the semantics is preserved. However, we may also want to make
changes in the specifications that do not preserve semantics. In these cases, the basic laws
proposed next are not suitable but this is trivially done by the user. Furthermore, it is a
good practice to show that the set of laws that we proposed are complete in some sense.
The standard approach is to show that the basic set of laws is sufficient to transform any
arbitrary model into a normal form expressed in terms of a small subset of the language
operators [Hoare et al., 1987]. This, however, is beyond the scope of this article.

A similar work has been done for Refinement Object Oriented Language (ROOL)
[Borba and Sampaio, 2000]. However, ROOL is less powerful for specifying structural
properties among types. Whereas ROOL supports only attribute declarations, as in Java
[Gosling et al., 1996], Alloy supports the declaration of bidirectional relations with arbi-
trary arities and multiplicities, as in UML with OCL. Another difference is that in ROOL
we can not define global constraints, such as those involving cardinality (number of in-
stances) of classes in the entire system. In Alloy and UML with OCL this is possible.

Related approaches [Lano and Bicarregui, 1998, Gogolla and Richters, 1998,
Sunyé et al., 2001, Evans, 1998] have proposed some transformations of UML models.
However some of the transformations do not completely preserve semantics. Further-
more, they are different from ours since they do not propose a small set of small-grain
transformations. They are much similar to refactorings, which are large-grain transforma-
tions that preserve semantics and may improve readability, design structure and reusabil-
ity.

Model transformations are useful because it is easier to transform models rather
than code. Usually, we only refactor code. Nevertheless, if we build a tool to relate code
and model transformations automatically, we only need to make changes in the models
which are simpler, and it automatically maps changes to the code.

The remainder of this paper is organized as follows. Section 2 overviews the
Alloy language. In Section 3, we present some basic laws for Alloy. The following
section illustrates how the laws presented in the previous section can be used to formally
refactor a design model for Java types. Finally, Section 5 discusses some related work
and presents our conclusions.

2. Alloy

Alloy is a strongly typed language that assumes a universe of elements partitioned into
subsets, each of which is associated with a basic type. An Alloy model or specification is a
sequence ofparagraphsof two kinds: signatures that are used for defining new types, and
formulas paragraphs, like facts and functions, used to record constraints. Each signature
has a set of objects (elements). These objects can be related by the relations declared in
the signatures.

A signature paragraph introduces a basic type – if it does not extend another signa-
ture – and a collection of relations, calledfields, along with the types of the fields and other
constraints on the values they relate. If a signature extends another, then its type is the
parent signature type. Besides signature extension, Alloy has other important structures
and operators like modules, polymorphism, commands for analyzing the specification,
that can be found elsewhere [Jackson et al., 2001].

Suppose that we want to model in Alloy part of a bank system where each bank
has a set of accounts and each account has an identifier that is unique. Figure 1 describes
the object model [Liskov and Guttag, 2001] of the system. Each box in an object model
represents a set of objects. The arrows are relations, sometimes called associations, and



Figure 1: Bank System Object Model

they indicate how objects of a set are related to objects in other sets. For instance, the
arrow labeledid from Account to String shows that each object fromAccount has a
field whose value object is aString object.

The multiplicity symbols are:! (exactly one),? (zero or one), * (zero or more) and
+ (one or more). Multiplicities annotations can appear on both ends of the arrow. If a
multiplicity symbol is omitted, * is assumed. The following piece of code introduces the
three signatures and two relations that models part of the bank system.

sig Bank {
accounts: set Account

}
sig Account {
id: String

}
sig String {}

In the field declaration ofBank, the keywordset specifies thataccounts maps
each element inBank to a set of elements inAccount. When we omit the keyword, as in
the declaration ofid, we specify that all accounts have one identifier (total function).

One of the formula paragraphs is called a fact. It is used to package formulas that
always hold, such as invariants about the elements of certain types and sets. The following
example introduces a fact namedRestrictions, establishing general properties about
the previously introduced signatures and relations.

fact Restrictions {
Account = Bank.accounts
all disj acc1,acc2:Account | (acc1.id) != (acc2.id)

}

The first formula states that there is no account that is not a bank account and
the other one states that different accounts have different identifiers. The keywordall is
the universal quantifier and we usedisj beforeacc1 andacc2 to indicate that they are
disjoint. In Alloy, the fact formulas are implicit declared as a conjunction of them.

Functions are parameterized formulas that can be applied by binding its parame-
ters to expressions whose types match the declared parameter types. They are especially
useful for specifying the behavior of the system operations. The subsequent example de-
clares a function that adds an account to a specific bank and returns a bank with the new
account added.

fun Bank::addAccount(acc:Account):Bank {
result.accounts = this.accounts + acc

}



The operator+ corresponds to the union operator. The keywordsthis and
result are anonymous variables that contain the receiver and result arguments. The
receiver and result arguments represent the pre and post state, respectively, of the
addAccount function.

The joinp.q of relationsp andq is the relation you get by taking every combination
of an element inp and an element inq, and including their join, if it exists. The relations
p andq may have any arity, as long as they are not both unary. The right type ofp must
match the left type ofq. This is a generalized definition of the standard join operator.
For instance, the join ofthis.accounts, wherethis is a bank andaccounts is a binary
relation that relates banks to accounts, returns the set of elements which are the relational
image ofthis in accounts. This expression yields all accounts of the receiver bank.

The semantics of an Alloy specification is all the possible assignments of values
to the signature names and field names that satisfy the implicit and explicit constraints
of the specification. In the previous example, it consists of all valid assignments to the
signature namesBank, Account andString and to the relation namesaccounts andid.
These assignments have to satisfy all implicit and explicit constraints of the specification.
For example, all elements in all assignments given toAccount have to be a bank account.

Models in Alloy are concise, analyzable, declarative and structural. None of these
features is new in itself. Alloy is different because it is declarative and analyzable at the
same time. It has been assumed for a long time that a high-level specification cannot be
analyzable and declarative at the same time. Functional programs are both declarative
and executable, but they do not have the same level of abstraction of Alloy’s specifica-
tions, which are more abstract. In Alloy, it is possible to automate different kinds of
simulations and analysis [Jackson et al., 2000, Jackson, 2000, Jackson and Fekete, 2001,
Jackson and Sullivan, 2000].

3. Basic Laws

In this section, we present some basic laws proposed for Alloy. These laws state properties
about signatures, formulas and relations. The first law establishes that we can always
introduce an empty signature declared with a fresh name. It indicates that we can also
remove an empty signature that is not used. We usedps to denote a set of paragraphs.
Since in Alloy we can not have two paragraphs with the same name then we have a
proviso stating that the name of the fresh signature does not appear inps.

Law 1 〈introduce empty signature〉

ps
=

ps
sig S {}

provided

(→) psdoes not declare any paragraph namedS;

(←) Sdoes not appear inps.

We write (→) before the proviso to indicate that this proviso is only required for
applications of this law from left to right. Similarly, we use (←) to indicate that it is only
for applying the law from right to left, and we use (↔) to indicate that the proviso is
necessary in both directions. It is important to notice that each basic law, when applied
from left to right or right to left, defines one transformation that preserve semantics.



We consider that two models in Alloy are equal if all possible assignments to
the names of one model are equivalent to the other model’s assignments in respect to all
names that are common to the models.

A similar law holds for facts, but as Alloy facts can not be referred from other
paragraphs, the elimination of an empty fact is always possible.

Law 2 〈introduce empty fact〉

ps
=

ps
fact F {}

provided

(→) psdoes not declare any paragraph namedF.

The next law establishes that we can add or remove a formula to a fact as long as
it can be deduced from the other formulas of the specification.

Law 3 〈introduce formula〉

ps
fact F {

forms
}

=

ps
fact F {

forms
f

}

provided

(↔) The formulaf can be deduced from the formulas inpsandforms.

Sincef is derived from other formulas, we guarantee that both specifications have
the same meaning. The constraints imposed by this formula are already imposed by the
others. We writeforms to denote a set of formulas. From predicate calculus, we infer
′A and B′ from the′A => B′ andA formulas. Therefore, we can apply this law. These laws
presented here focus only on Alloy structures. However, relational operator properties and
predicate calculus properties can also be applied to Alloy formulas. Another law indicates
that we can always move a formula from one fact to another.

Law 4 〈move formula〉

ps
fact F {

forms
f

}

fact G {

forms′

}

=

ps
fact F {

forms
}

fact G {

forms′

f
}

We can move a formula from one fact to another because in Alloy it does not
matter if one formula is declared in one fact or in another since the semantics of Alloy
specifications treat them as a conjunction of all fact formulas.

We can think about Laws 1 and 2 as a neutral element. Similarly, Law 3 states
the idempotence property of the conjunction logical operator, whereas Law 4 states the
associative property of the conjunction logical operator.



Besides those laws for dealing with facts, we also have laws for manipulating
relations. The next law states that we can introduce a new relation along with its definition,
that is a formula in the formr = exp, establishing a value for the relation. We can also
remove a relation that is not used.

Law 5 〈introduce relation and its definition〉

ps
sig S {

rs
}

fact F {

forms
}

=

ps
sig S {

rs,
r : set T

}

fact F {

forms
r=exp

}

provided

(→) The family ofS does not declare any relation namedr in ps, andT is eitherS or a
signature name declared inps;

(←) Ther relation does not appear inpsandforms.

The family of theS signature is the set of all signatures that extend and are ex-
tended byS including itself. The precondition stating that the fresh relation name does
not appear in theSsignature family guarantees that we do not have conflict names. Alloy
does not allow two relations with the same name in a family signature.

Notice that the previous law can be used to simply introduce a relation, without
any definition. We have just to takeexpas beingr itself, introducing a tautology, which
can be trivially eliminated with Law 3.

A relation qualified as asetof T, declared in theSsignature, indicates that every
element inS can be related to any number ofT elements. Since it does not impose any
constraint in the mapping, we assure that the previous law preserves the implicit and
explicit constraints. Therefore, we do not introduce an inconsistency. The type of ther
relation can also be qualified asscalar or option of T. The first qualifier defines a total
function and the last one a partial function fromS to T, in this case. The default qualifier
is scalar. Due to the implicit constraint introduced by thescalar andoption qualifiers,
we can not introduce a relation declared with these qualifiers since the implicit constraints
can contradict previous implicit and explicit constraints of the specification.

In this context, we can only introduce a relation with thesetqualifier. Neverthe-
less, if we derive from the implicit and explicit constraints that this new relation is a total
or a partial function then we can apply Law 3 from left to right and introduce this formula.
After that, we can change the type of the relation by applying the laws that we show next.
Hereafter we user instead ofS$r because we assume, for simplicity, that there is no other
relation with the same name in the specification. The notationS$r alternatively denotes a
reference to ther relation of theSsignature.

The following two laws allow us to change the type of the relations fromset to
scalar or option. The first of them establishes that we can change the type of a relation
from setto scalar, or vice-versa, ifr is a total function.



Law 6 〈change relation type: from set to scalar〉

ps
sig S {

rs,
r : set T

}

fact F {

forms
all s : S | one s.r

}

=

ps
sig S {

rs,
r : scalar T

}

fact F {

forms
}

A similar law is proposed to change the type of a relation fromset to option or
vice-versa. In this case, we need a formula that establishes thatr is a partial function.

Law 7 〈change relation type: from set to option〉

ps
sig S {

rs,
r : set T

}

fact F {

forms
all s : S | sole s.r

}

=

ps
sig S {

rs,
r : option T

}

fact F {

forms
}

It is important to notice that Laws 5, 6 and 7 are also valid if theS signature
extends other signatures. We have proved the soundness of these laws and proposed other
laws, such as pull up relation, but these are beyond the scope of this article. We proved
the soundness of these laws based on the denotational Alloy’s semantics that we propose
using Alloy. We prefer to propose small-grain transformations because it is easier to prove
that they preserve semantics. Composing them, we can derive large-grain transformations,
which consequently preserve semantics. Examples of the use of the laws can be found in
the following section.

4. Refactoring Java Types Specification

Maintaining an object-oriented program often requires structural changes, such as chang-
ing the relations between classes. This is useful for software evolution, when designers
are interested in restructuring the specification for improvement, or to introduce new re-
quirements in a more adequate way.

In this section, we show how we can refactor a simple but non-trivial Alloy speci-
fication by simply using the presented laws. This illustrates that the laws can be useful to
refactor models. We consider here a specification that models part of a Java type-checker
specification. This model was written by Daniel Jackson1, with a few syntax differences,
and is part of the Alloy Analyzer distribution package2.

4.1. Initial Specification

The specification of Java types describes the basic notions of typing in Java as depicted in
Figure 2. An arrow with a closed head form, like fromObject to Class, denotes a subset

1http://sdg.lcs.mit.edu/˜dnj
2http://alloy.mit.edu



Figure 2: Object Model of Daniel Jackson’s Java Types

relationship. In this case,Object is a subset ofClass. If two subsets share an arrow, they
are disjoint. For instance,Class andInterface are disjoint. If the arrowhead is filled,
the subsets exhaust the superset, so there are no members of the superset that are not
members of one of the subsets. In this case, the subsets form apartition: every member
of the superset belongs to exactly one subset.

The! notation in theObject box indicates that the set this box represents contains
one element. This model ignores primitive types andnull references. EachType may
have a set of subtypes. Types are partitioned intoClass andInterface types andObject
is a particular class. The following Alloy’s fragment describes the object model in Figure
2.

sig Type {
subtypes: set Type

}
sig Class, Interface extends Type {}

In Alloy, we can declare several signatures at once if they do not declare any
relation, as showed in the declaration ofClass andInterface. One signature can extend
another one establishing that the extended signature is a subset of the parent signature.
This does not introduce another basic type, as mentioned before. The extended signature
has the same type of the parent signature.

static sig Object extends Class {}

Thestatic keyword denotes that the signature has exactly one element. In this
example, theObject signature has only one element.

The following fact declaration states that every type is subtype ofObject but is
not a subtype of itself. Additionally, it specifies that every type is a subtype of at most one
class. It also declares that there exists no type that is both class and interface and any type
is exclusively a class or an interface. The operators∗, ˆ and ˜ correspond, respectively, to
reflexive transitive closure, transitive closure and transpose of a relation.

fact TypeHierarchy {
Type in (Object.*subtypes)
no t: Type | t in (t.ˆsubtypes)
all t: Type | sole ((t.˜subtypes) & Class)



no (Class & Interface)
Type = (Class + Interface)

}

The operator& corresponds to the intersection operator. The keywordno is a
quantifier in the second formula. But, when applied to an expression, as in the fourth
formula, defines a predicate that there is no element in the expression. The keywordin

in Alloy can denote a subset operator, like in the first formula of the previous fact, and a
membership operator as in the second formula declared before. The keywordsole, when
applied to an expression, defines a predicate. For instance,sole (t. subtypes & Class)
is true if the expression(t. subtypes & Class) has at most one element.

So far, we have specified types in Java. Now we describe part of a simple type
checker of some expressions, for simplicity. The following signature declarations express
that every object has a type (its creation type) that is a class. A variable may hold an
instance, and has a declared type.

sig Instance {
type: Class

}
sig Variable {
holds: option Instance,
type: Type

}

Finally, theTypeSoundness fact states that all instances held by a variable have
types that are direct or indirect subtypes of the variable’s declared type.

fact TypeSoundness {
all v: Variable | (v.holds.type) in (v.type.*subtypes)

}

4.2. Refactored Specification

The previous model describes Java types in terms of thesubtypes relation. However, we
want to refactor this model to describe Java types in terms of supertype relations such as
extends3 andimplements. In order to guarantee that the resulting model is equivalent to
the original one, we should not proceed in an ad hoc way, making arbitrary changes to the
model. In fact, we can use Laws 2, 3, 5 and 7 to change the specification in a step-by-step
way, as illustrated next, preserving its semantics. The desired object model is described
in Figure 3.

To simplify the following refactoring steps, we consider thatps contains the sig-
natures and facts declared before, except theType signature. At the end of the section,
we show the complete specification. In order to introduce the new relations, we can use
Law 5, but as this law requires a fact, we first apply Law 2 from left to right to introduce
an empty fact.

Next, we introduce, using Law 5 from left to right, the relationsextends and
implements, in theType signature, and their definitions, in theDefinitions fact.

ps
sig Type {
subtypes: set Type,
extends: set Class,

3It is important to observe that no relation in Alloy can be named extends because it is a keyword.
Nevertheless, we use it here only to improve readability.



Figure 3: Object Model of Java Types Refactored

implements: set Interface
}
fact Definitions {
implements = (˜subtypes & (Type->Interface))
extends = (˜subtypes & (Type->Class))

}

The notation− > represents the product operator that combines every element in
the left operand with every element in the right operand and concatenates them. These
elements can also be a tuple of elements. When applied to sets, this operator represents
the standard Cartesian product. Notice that we can apply the law because noextends

andimplements relations are declared in theType’s family. Observe also that the types
of the new relations are valid in the initial specification.

Our aim is to derive thesubtypes definition in order to replace it by its defini-
tion and eventually remove it from the specification. To derive its definition, notice that
extends andimplements have the same type (Type − > Type). We use some predicate
calculus and set theory properties [Spivey, 1989] and some formulas in the specification,
within brackets, to justify every step in the derivation. Since they have the same type, we
can derive that their union is:

(extends + implements =
(˜subtypes & (Type->Class)) + (˜subtypes & (Type->Interface)))
[A&B + A&C = A&(B+C)]

<=> (extends + implements =
˜subtypes & ((Type->Class) + (Type->Interface)))

[(A->B) + (A->C) = (A->(B+C))]
<=> (extends + implements =

˜subtypes & (Type->(Class + Interface)))
[Type = (Class+Interface)]

<=> (extends + implements = ˜subtypes & (Type->Type))
<=> (extends + implements = ˜subtypes) [(A=B) => (˜A=˜B)]
<=> (˜(extends + implements) = ˜˜subtypes) [˜(A+B) = (˜A+˜B)]
<=> (˜extends + ˜implements = ˜˜subtypes) [˜˜A = A]
<=> (˜extends + ˜implements = subtypes) [(A=B) => (B=A)]
<=> (subtypes = ˜extends + ˜implements)



Since this formula is deduced fromDefinitions, using Law 3 from left to right,
we can introduce this formula in the specification.

ps
sig Type {
subtypes: set Type,
extends: set Class,
implements: set Interface

}
fact Definitions {
implements = (˜subtypes & (Type->Interface))
extends = (˜subtypes & (Type->Class))
subtypes = (˜extends + ˜implements)

}

From the third formula of the TypeHierarchy fact, we can deduce that
all t : Type | sole (t.extends), as explained next.

all t: Type | sole ((t.˜subtypes) & Class)
[subtypes = ˜extends + ˜implements]

<=> all t: Type | sole ((t.˜(˜extends + ˜implements)) & Class)
[˜(A+B) = ˜A + ˜B]

<=> all t: Type | sole ((t.(˜˜extends + ˜˜implements)) & Class)
[r.(A+B) = r.A + r.B]

<=> all t: Type | sole ((t.˜˜extends + t.˜˜implements) & Class)
[˜˜A = A]

<=> all t: Type | sole ((t.extends + t.implements) & Class)
[(A+B)&C = A&C + B&C]

<=> all t: Type | sole (((t.extends) & Class) +
((t.implements) & Class)) [no (Class & Interface)]

<=> all t: Type | sole (((t.extends) & Class) + {}) [A + {} = A]
<=> all t: Type | sole ((t.extends) & Class)
<=> all t: Type | sole (t.extends)

Then we can apply Law 3 from left to right introducing
all t : Type | sole (t.extends) in the Definitions fact since we derive a for-
mula stating thatextends is a partial function.

ps
sig Type {
subtypes: set Type,
extends: set Class,
implements: set Interface

}
fact Definitions {
implements = (˜subtypes & (Type->Interface))
extends = (˜subtypes & (Type->Class))
subtypes = (˜extends + ˜implements)
all t:Type | sole (t.extends)

}

Now we can change the type of theextends relation fromset tooption applying
Law 7 from left to right.

ps
sig Type {
subtypes: set Type,



extends: option Class,
implements: set Interface

}
fact Definitions {
implements = (˜subtypes & (Type->Interface))
extends = (˜subtypes & (Type->Class))
subtypes = (˜extends + ˜implements)

}

Next, we replace the references to thesubtypes relation by its definition. No-
tice that from every formula, except its definition, containing thesubtypes relation, we
can derive a formula replacing thesubtypes relation by its definition and insert it to the
specification applying Law 3 from left to right. Consequently, these new formulas can
also derive the formulas withsubtypes. Next, we can remove all formulas that con-
tain subtypes from the specification applying Law 3 from right to left. The notation
ps (subtypes = ˜extends+˜implements) denotes that we replace every occurrence of
subtypes in ps by ˜extends+˜implements.

ps (subtypes = ˜extends + ˜implements)
sig Type {
subtypes: set Type,
extends: option Class,
implements: set Interface

}
fact Definitions {
implements = (˜(˜extends + ˜implements) & (Type->Interface))
extends = (˜(˜extends + ˜implements) & (Type->Class))
subtypes = (˜extends + ˜implements)

}

The resulting first and second formulas of theDefinitions fact are trivially valid
formulas. We can deduce using some set theory properties and formulas in the specifica-
tion, thatimplements = implements andextends = extends, respectively. Therefore,
we can remove them applying Law 3 from right to left.

ps (subtypes = ˜extends + ˜implements)
sig Type {
subtypes: set Type,
extends: option Class,
implements: set Interface

}
fact Definitions {
subtypes = (˜extends + ˜implements)

}

Sincesubtypes does not appear inps, because we replaced it by its definition, we
can remove this relation and its definition using Law 5 from right to left. After that, the
Definitions fact becomes empty and can be removed applying Law 2 from right to left.
This is the last step for replacing thesubtypes relation by theextends andimplements
relations. The resulting specification is shown next:

sig Type {
extends: option Class,
implements: set Interface

}
part sig Class, Interface extends Type {}



static sig Object extends Class {}

fact TypeHierarchy {
Type in (Object.*(˜extends + ˜implements))
no t: Type | t in (t.ˆ(˜extends + ˜implements))
all t: Type | sole ((t.˜(˜extends + ˜implements)) & Class)
no (Class & Interface)
Type = (Class + Interface)

}

sig Instance {
type: Class

}
sig Variable {
holds: option Instance,
type: Type

}

fact TypeSoundness {
all v: Variable |
(v.holds.type) in (v.type.*(˜extends + ˜implements))

}

Now the specification models the Java types in terms of supertype relation. How-
ever, the resulting model present some formulas that are not so readable, like some for-
mulas in theTypeHierarchy fact. We can replace them by other formulas. For instance,
we can replace the third formula of this fact byall t : Type | sole (t.extends) us-
ing some set theory and predicate calculus properties and formulas of the specification.
In fact, we can even remove this formula. First, applying Law 7 from right to left
derives the same formula. Next, using Law 3 from right to left we can remove one
of them. Finally, applying Law 7 from left to right, result a specification without the
all t : Type | sole (t.extends) formula.

We have refactored other specifications and have derived other refactorings such
as introducing a generalization and moving a relation. However, due to the lack of space,
we will not show them here.

5. Conclusions

In this paper we propose basic laws for Alloy. These laws are important not only to define
the axiomatic semantics of the modeling language but also to refactor specifications, as
illustrated for the Java type’s model.

The laws presented here have been proved to be sound with respect to a formal
semantics for Alloy. Consequently, they should serve as tools for carrying out model
transformations. One immediate application of the basic laws is to define an interface
from which one can derive more elaborate laws such as refactorings. As illustrated in
Section 4, they can also be used to refactor designs in a step-by-step way. We do not need
to prove that the initial and final Java types specifications have the same semantics because
the basic laws, that we used, guarantees that each one preserve semantics. Therefore, the
composition of them also preserves semantics.

The main focus of the basic laws is the architects. The architects will use them to
derive refactorings (large-grain transformations) that will be available for the designers.



The designers will use only the refactorings derived by the architects. We have deduced
some refactorings and apply three of them to a bank system’s specification. In this case,
the designer only needs to apply three refactorings rather than refactoring the bank sys-
tem’s specification in around twenty steps applying the basic laws, as we did in the Java
type’s specification. Formalizing further basic laws, like one stating in which conditions
we can introduce a generalization, refactorings, and applying them to other realistic case
studies are some topics for further research.

Most of the basic laws are very simple to apply since their pre-conditions are
simple syntactic conditions. However, some laws like the one to introduce a formula (Law
3) might require, in order to avoid errors and to help the architect, the use of an automated
theorem prover or a proof assistant to verify if two formulas are equivalent, for example.
Alloy has can also be used in these cases. However, we can only guarantee in the Alloy
Analyzer, that the assertions are valid for a pre-defined scope. We extended the Alloy
Analyzer tool to include the implementation of most of the basic laws presented here,
where the user does not need to verify the pre-conditions and apply them. The user only
requires informing the parameter values of the transformations. Another approach that we
can use to automate the implementation of them is to use the Maude [Meseguer, 1993]
term rewriting system.

Although we already have a set of basic laws for Alloy, some of them we show
here, we still need to prove a reduction theorem stating that our set of laws is complete
in the sense of allowing reduction of arbitrary Alloy specifications to a normal form ex-
pressed in a small subset of the language operators following approaches adopted for
ROOL [Borba et al., 2003], for imperative languages [Hoare et al., 1987], among oth-
ers. Another further research topic is to check the relation between modeling laws and
programming laws; for instance, the laws of ROOL. In particular, we need to investigate
if the design refactorings have corresponding code refactorings.

Related work [Sunyé et al., 2001, Evans, 1998, Gogolla and Richters, 1998,
Lano and Bicarregui, 1998] has been carried out on transformation of UML design mod-
els. However, some transformations [Sunyé et al., 2001, Gogolla and Richters, 1998,
Lano and Bicarregui, 1998] do not preserve semantics. For instance, it is argued that
creating a generalization between classes preserve semantics. However, the constraints in
the specification can become inconsistent by introducing a generalization. For instance,
we can not extend theB class with theA class when we have a explicit constraint in the
specification stating thatB has more elements thanA. The introduction of a generalization
in this case will make the specification inconsistent since the generalization introduces an
implicit constraint thatA containsB. Therefore, we deduce, from this implicit constraint,
that A has the same elements or more elements thanB, which contradicts the explicit
constraint in the specification.

In addition, another work [Lano and Bicarregui, 1998] states that logically
strengthening a class invariant is a refinement transformation, although adding a formula
can sometimes turn the specification inconsistent. Another work [Evans, 1998] removes
classes and relations without checking if there are some constraints about them in the
specification, which is not acceptable in a formal setting. These transformations do not
preserve semantics because some of them use an UML semantics that is ambiguous. Oth-
ers define part of UML semantics but do not verify the transformations soundness or do
not consider OCL constraints.

